Appendix: SAS® program to allocate subjects to two treatments in blocks of randomly varying sizes of 4 and 6

/* Macro to perform Block Randomization */

%macro randblk(trial,centre,ngroups,block);

   /*--------------------------------------------------- */

   /* Trial  : Trial number                              */

   /* Centre : Centre number -> Seed is Trial * Centre   */

   /* ngroups: is the number of groups for randomization */

   /* Block  : the size of the block size                */

   /*--------------------------------------------------- */

   /* block to hold treatment groups in calling program  */

   /* array blk (*)  blk01-blknn;                        */

   /*--------------------------------------------------- */

   rgrp = 1;

   pos = 1;

   rseed = &trial*&centre;

   /*--------------------------------------------------------*/

   /* Initialize block with equal number of treatment groups */

   /*--------------------------------------------------------*/

   do pos = 1 to &block;

      blk(pos) = rgrp;

      if rgrp = &ngroups then rgrp = 1;

      else rgrp = rgrp + 1;

      end;

   /*-------------------------------------------------------*/

   /* Pseudo-random permutation of treatment groups         */

   /*-------------------------------------------------------*/

   do ri = 1 to &block;

      rperm = min(int(&block*ranuni(rseed))+1,&block);

      rtemp = blk(ri);

      blk(ri) = blk(rperm);

      blk(rperm) = rtemp;

      end;

   %mend randblk;

Proc format;

   value grand 1 = 'Misoprostol 600mg' 2 = 'Control';

Title1 'Project 97901 - Misoprostol Trial';

Title2 'Allocation of subjects in blocks of sizes 4 and 6, centre X';

Data Rand;

   array blk     (*) blk1-blk6;

   array blksize (*) bls1 bls2;

   bls1 = 4;          /* block size randomly changed from 4 to 6 */

   bls2 = 6;

   trial = 97901;     /* Trial Number */

   centre = 0896;    /* Centre X */

   from_sub = 1;

   to_sub = 700;

   keep trial centre subj chk groupn;

   soffset = 0;

   do while (soffset < to_sub);

      rsize = min(int(2*ranuni(rseed))+1,2);

      %randblk(trial,centre,2,blksize(rsize));

      subj  = soffset;

      do si = 1 to blksize(rsize) while (subj < to_sub);

         subj = soffset + si;

         groupn = blk(si);

         output;

         end;

      soffset = soffset + blksize(rsize);

      end;

   stop;

